

المحاضرة التاسعة

استخدام المؤشر Cursor بطريقة For – Loops

يوفر لنا الأوراكل طريقة أخرى للتعامل مع ال **Cursor** وذلك باستخدام حلقة **For – Loop** لتسهيل التعامل مع المؤشر حيث أن حلقة **For** تقوم بثلاث مهام في آن واحد وهي :-

- (1) - **Open the cursor.**
- (2) - **Fetch data from the cursor.**
- (3) - **Close the cursor.**

Syntax: -

```
FOR for_variable IN cursor_name LOOP
    ...
    ...
END LOOP;
```

مهم جداً:-

المتغير *for_variable* هو أي متغير نكتبه ويقوم الأوراكل تلقائياً بتعريفه ويكون في شكل سجل Record مساوياً للحقول التي اخترناها في جملة **Select** عند تعريف المؤشر .
وعندما نريد الإشارة إلى أحد الحقول الداخلية في السجل نكتب اسم المتغير ثم نقطة ثم اسم الحقل المراد.



يجب الانتباه إلى أننا لا يمكن أن نشير إلى المتغير *for_variable* خارج الحلقة **For**.

مثال: استخدام For – Loop :-

```
SQL> DECLARE
2  CURSOR my_for_cursor is
3  select empno,ename,job from emp
4  where sal >=1500;
5  BEGIN
6  dbms_output.enable;
7  For mem_rec IN my_for_cursor loop
```

```
8 dbms_output.put_line(mem_rec.ename||' : '||mem_rec.job);
9 END loop;
10 END;
11 /
```

```
ALLEN : SALESMAN
JONES : MANAGER
BLAKE : MANAGER
CLARK : MANAGER
SCOTT : ANALYST
KING : PRESIDENT
TURNER : SALESMAN
FORD : ANALYST
```

PL/SQL procedure successfully completed.

مثال: استخدام متغيرات مع Cursor – For – Loop :-

```
SQL> DECLARE
2 my_sal emp.sal%type;
3 CURSOR my_for_cursor(sal_value number) is
4 select empno,ename,job from emp
5 where sal >= sal_value;
6 BEGIN
7 dbms_output.enable;
8 my_sal := 1500;
9 For mem_rec IN my_for_cursor(my_sal) loop
10 dbms_output.put_line(mem_rec.ename||' : '||mem_rec.job);
11 END loop;
12 END;
13 /
```

```
ALLEN : SALESMAN
JONES : MANAGER
BLAKE : MANAGER
CLARK : MANAGER
SCOTT : ANALYST
KING : PRESIDENT
TURNER : SALESMAN
FORD : ANALYST
```

PL/SQL procedure successfully completed.

مثال: الأوراق يعطي رسالة خطأ إذا تعاملنا مع ال Cursor خارج الحلقة :-

```
SQL> DECLARE
2 my_sal emp.sal%type;
3 CURSOR my_for_cursor(sal_value number) is
4 select empno,ename,job from emp
5 where sal >= sal_value;
6 BEGIN
7 dbms_output.enable;
8 my_sal := 1500;
9 For mem_rec IN my_for_cursor(my_sal) loop
10 dbms_output.put_line(mem_rec.ename||' : '||mem_rec.job);
11 end loop;
12 dbms_output.put_line('The total no of employee is: '||my_for_cursor%rowcount);
13 end;
14 /
```

```
ALLEN : SALESMAN
JONES : MANAGER
BLAKE : MANAGER
CLARK : MANAGER
SCOTT : ANALYST
KING : PRESIDENT
TURNER : SALESMAN
FORD : ANALYST
DECLARE
*
```

```
ERROR at line 1:
ORA-01001: invalid cursor
ORA-06512: at line 12
```

في المثال السابق حاولنا استخدام **%rowcount** خارج حلقة **For – Loop** وذلك لمعرفة عدد السجلات المسترجعة ولكن الأوراق قام بإعطاء رسالة خطأ لأن حلقة **For** تقوم بإغلاق المؤشر بمجرد انتهاء الحلقة.

ولإصلاح الخطأ السابق أنظر المثال في التالي.

مثال: إصلاح الخطاء الوارد في المثال السابق :-

```
SQL> DECLARE
2 my_sal emp.sal%type;
3 my_counter number;
4 CURSOR my_for_cursor(sal_value number) is
5 select empno,ename,job from emp
6 where sal >= sal_value;
7 BEGIN
8 dbms_output.enable;
9 my_sal := 1500;
10 For mem_rec IN my_for_cursor(my_sal) loop
11 dbms_output.put_line(mem_rec.ename||' : '||mem_rec.job);
12 my_counter := my_for_cursor%rowcount;
13 end loop;
14 dbms_output.put_line('The total no of employee is: '||my_counter);
15 end;
16 /
```

ALLEN : SALESMAN
JONES : MANAGER
BLAKE : MANAGER
CLARK : MANAGER
SCOTT : ANALYST
KING : PRESIDENT
TURNER : SALESMAN
FORD : ANALYST

The total no of employee is: 8

PL/SQL procedure successfully completed.

لإصلاح الخطاء في المثال السابق قمنا بتعريف متغير ذاكرة **my_counter** وقمنا بإسناد قيمة **%rowcount** له داخل حلقة **For – Loop** ثم استخدمناه بعد ذلك خارج الحلقة.

النوع الثاني: المؤشر الضمني IMPLICIT CURSOR

وهو مؤشر **Cursor** يقوم الأوراكل بإنشائه وفتحه تلقائياً لكل جملة من جمل **SQL** لا تكون جزء من المؤشر الضمني **Explicit Cursor** . ولا يمكننا أن نستخدم الخيارات **Open, Fetch, Close** مع المؤشر الضمني وذلك لأن ليس له اسم لذلك نستخدم له اسم **sql** عندما نحاول الإشارة إلى إحدى خصائصها مثل أن نكتب **sql%rowcount**.
ويستخدم المؤشر الضمني **Implicit Cursor** عادة مع أوامر **Update** و **Delete** كما سيتضح بالمثال التالي.

مثال: استخدام ال Implicit Cursor :-

```
SQL> BEGIN
2 dbms_output.enable;
3 update emp
4 set sal = 5000
5 where job = 'NO JOB';
6 if sql%notfound then
7 dbms_output.put_line('No Employee Found ???');
8 else
9 commit;
10 end if;
11 END;
12 /
```

No Employee Found ???

PL/SQL procedure successfully completed.

مثال: استخدام %rowcount مع ال Implicit Cursor :-

```
SQL> BEGIN
2  dbms_output.enable;
3  Update my_emp
4  set sal = sal + 100
5  where
6  job = 'SALESMAN';
7  dbms_output.put_line('The total record updated is: '||sql%rowcount);
8  end;
9  /
```

The total record updated is: 4

PL/SQL procedure successfully completed.

	ملاحظة مهمة: -
<input checked="" type="checkbox"/>	لا يمكننا استخدام %isopen مع المؤشر الضمني Implicit Cursor وذلك لأن الأوراكل يقوم بإغلاق المؤشر Cursor بمجرد الانتهاء من تنفيذ جملة SQL.